Abhängige Berechnungen

Der Wert eines Ausdrucks wird berechnet und ausgegeben:

```
ATTR value: int;

RULE: Root ::=  Expr  COMPUTE
   printf ("value is %d\n", Expr.value);
END;

TERM Number: int;

RULE: Expr ::= Number COMPUTE
  Expr.value = Number;
END;

RULE: Expr ::= Expr Opr Expr COMPUTE
  Expr[1].value = Opr.value;
  Opr.left  = Expr[2].value;
  Opr.right = Expr[3].value;
END;

SYMBOL Opr: left, right: int;

RULE: Opr ::=  '+'  COMPUTE
  Opr.value  = ADD (Opr.left, Opr.right);
END;

RULE: Opr ::=  '*'  COMPUTE
  Opr.value = MUL (Opr.left, Opr.right);
END;
```

Praktikum Sprachimplementierung mit Werkzeugen WS 1999/ 2000 / Folie 301

Ziele:

Berechnungen definieren und benutzen Attribute

im Vorlesungsteil:

Vollständigkeit und Konsistenz der Attributberechnungen

nachlesen:

Computation in Trees: 2.1

Verständnisfragen:

- Fügen Sie eine Berechnung hinzu, so daß die Konsistenz verletzt wird.
- Welche Berechnungen könnte man zufügen oder weglassen, ohne daß Vollständigkeit oder Konsistenz verletzt wuerden?
- Welche Wirkung hat das Vertauschen der Reihenfolge von Berechnungen?

Vor- und Nachbedingungen von Berechnungen

Ausgabe von Ausdrücken in Postfix-Form:

```
RULE: Root ::= Expr COMPUTE
  Expr.print = "yes";
  printf ("\n") <- Expr.printed;
END;

RULE: Expr ::= Number COMPUTE
  Expr.printed =
    printf ("%d ", Number) <- Expr.print;
END;

RULE: Opr  ::= '+' COMPUTE
  Opr.printed = printf ("+ ") <- Opr.print;
END;

RULE: Opr  ::= '*' COMPUTE
  Opr.printed = printf ("* ") <- Opr.print;
END;

RULE: Expr  ::= Expr Opr Expr COMPUTE
  Expr[2].print = Expr[1].print;
  Expr[3].print = Expr[2].printed;
  Opr.print = Expr[3].printed;
  Expr[1].printed = Opr.printed;
END;
```

Die Attribute `print` und `printed` haben keinen Wert `(Typ VOID)`.

Sie beschreiben Zustände als Vor- und Nachbedingungen für Berechnungen:

`Expr.print:` Postfix-Ausgabe bis vor diesem Knoten ist erledigt.

`Expr.printed:` Postfix-Ausgabe bis einschließlich dieses Knotens ist erledigt.

Praktikum Sprachimplementierung mit Werkzeugen WS 1999/ 2000 / Folie 302

Ziele:

Zustandsattribute fuer Reihenfolge-Abhängigkeiten

im Vorlesungsteil:

Benutzung von Zustandsattributen

nachlesen:

Computation in Trees: 2.2

Verständnisfragen:

Welche Ausgabe würde erzeugt, wenn man die Zustandsattribute und Abhängigkeiten davon weglassen wuerde?

Abhängigkeitsmuster: CHAIN

CHAIN spezifiziert **left-right-depth-first** Abhängigkeit. Berechnungen werden in die CHAIN „eingehängt“.

Ausgabe von Ausdrücken in Postfix-Form:

```
CHAIN print: VOID;

RULE: Root ::= Expr COMPUTE
  CHAINSTART HEAD.print = "yes";
  printf ("\n ") <- TAIL.print;
END;

RULE: Expr ::= Number COMPUTE
  Expr.print =
    printf ("%d ", Number) <- Expr.print;
END;

RULE: Opr ::= '+' COMPUTE
  Opr.post = printf ("+") <- Opr.pre;
END;

RULE: Expr ::= Expr Opr Expr COMPUTE
  Opr.pre = Expr[3].print;
  Expr[1].print = Opr.post;
END;
```

Wertberechnung entlang einer CHAIN, z. B. Variablenadressen:

```
CHAIN RelAdr: int;

RULE: Block ::= '{' Sequence '}' COMPUTE
  CHAINSTART HEAD.RelAdr = 0;
END;

RULE: Definition ::= 'define' Ident COMPUTE
  Definition.RelAdr =
    ADD (Definition.RelAdr, VariableSize);
END;
```

Die Kontexte, in denen die CHAIN nicht benutzt wird, brauchen nicht erwähnt zu werden!

Praktikum Sprachimplementierung mit Werkzeugen WS 1999/ 2000 / Folie 303

Ziele:

Benutzung des CHAIN-Konstruktes

im Vorlesungsteil:

Vergleich mit voriger Folie

nachlesen:

Computation in Trees: 3.3

Übungsaufgaben:

Überpruefen Sie mit Eli durch Ableiten von :ExpInfo, welche Berechnungen Liga fuer diese CHAIN-Benutzung ergänzt.

Verständnisfragen:

Eine CHAIN-Benutzung wie Expr.print bedeutet je nach Auftreten Unterschiedliches. Erläutern Sie.

Abhängigkeitsmuster: INCLUDING

Ein Attribut wird tiefer im Unterbaum benutzt.

`INCLUDING Block.depth` bezieht sich auf das `depth`-Attribut des nächsten Oberknotens `Block`.

Berechnung der Schachtelungstiefe von Blöcken:

```
ATTR depth: int;

RULE: Root ::= Block COMPUTE
  Block.depth = 0;
END;

RULE: Statement ::= Block COMPUTE
  Block.depth = ADD (INCLUDING Block.depth, 1);
END;

TERM Ident: int;

RULE: Definition ::= ‘define' Ident  COMPUTE
  printf ("%s defined on depth %d\n ",
           StringTable (Ident),
           INCLUDING Block.depth);
END;
```

Die Kontexte zwischen der Definition und der Benutzung der `INCLUDING`-Attribute brauchen nicht erwähnt zu werden!

Praktikum Sprachimplementierung mit Werkzeugen WS 1999/ 2000 / Folie 304

Ziele:

Benutzung des INCLUDING-Konstruktes

im Vorlesungsteil:

Typische Anwendungen

nachlesen:

Computation in Trees: 3.1

Übungsaufgaben:

Welche Attribute und Berechnungen müßten Sie einfügen, wenn Sie das INCLUDING-Konstrukt ersetzen wollten? Überprüfen Sie dies mit Eli durch Ableiten von :ExpInfo.

Abhängigkeitsmuster: CONSTITUENTS für Zustände

Zusammenfassen von Attributen aus Unterbaum

`CONSTITUENTS Definition.DefDone` bezieht sich auf die `DefDone`-Attribute aller `Definition`-Knoten im Unterbaum.

Ausgabe aller Definitionen vor allen Anwendungen:

```
RULE: Block ::= '{' Sequence '}' COMPUTE
  Block.DefDone =
    CONSTITUENTS Definition.DefDone;
END;
RULE: Definition ::= 'Define' Ident COMPUTE
  Definition.DefDone =
    printf ("%s defined in line %d\n",
            StringTable(Ident), LINE);
END;
RULE: Usage ::= 'use' Ident COMPUTE
   printf ("%s used in line %d\n ",
           StringTable(Ident), LINE),
   <- INCLUDING BLOCK.DefDone;
END;
```

Die Kontexte zwischen der Definition und der Benutzung der `CONSTITUENTS`-Attribute brauchen nicht erwähnt zu werden!

Praktikum Sprachimplementierung mit Werkzeugen WS 1999/ 2000 / Folie 305

Ziele:

Benutzung des CONSTITUENTS-Konstruktes

im Vorlesungsteil:

- Typische Anwendungen
- Kombination mit INCLUDING

nachlesen:

Computation in Trees: 3.2

Übungsaufgaben:

Welche Attribute und Berechnungen müßten Sie einfügen, wenn Sie das CONSTITUENTS-Konstrukt ersetzen wollten? Überprüfen Sie dies mit Eli durch Ableiten von :ExpInfo.

Abhängigkeitsmuster: CONSTITUENTS für Werte

Zusammenfassung von Werten erfordert

```
CONSTITUENTS ... WITH (Type, BinFct, UnFct, ZeroFct)
```

`Type`	Typ des Ergebnis und der propagierten Werte
`BinFct`	verknüpft zwei Werte; sollte assoziativ sein
`UnFct`	erzeugt einen Wert aus dem Attributwert
`ZeroFct`	erzeugt einen "neutralen" Wert

Ausgabe der Anzahl der Anwendungen:

```
ATTR Count: int;

RULE: Block ::= '{' Sequence '}' COMPUTE
  printf  ("%d uses occurred\n",
           CONSTITUENTS Usage.Count
           WITH (int, ADD, IDENTICAL, ZERO));
END;

RULE: Usage ::= 'use' Ident COMPUTE
  Usage.Count = 1;
END;
```

Die Werte werden gemäß der Baumstruktur zusammengefaßt.

Praktikum Sprachimplementierung mit Werkzeugen WS 1999/ 2000 / Folie 306

Ziele:

Benutzung des CONSTITUENTS-Konstruktes fuer Werte

im Vorlesungsteil:

Typische Anwendungen

nachlesen:

Computation in Trees: 3.2

Übungsaufgaben:

Schreiben Sie fuer einen bestimmten Baum den Term mit allen Aufrufen der Funktionen ADD, IDENTICAL und ZERO auf, mit dem das CONSTITUENTS-Konstrukt berechnet wird.

Verständnisfragen:

Warum sollte die BinFct assoziativ sein und die ZeroFct einen bezüglich der BinFct neutralen Wert liefern?